Solutions to text problems

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

Exercise 4.1 (Source notebook: RollBasicAndGeneralized)

Here is a function to build a table of all possible n successive realizations of q.

```
qTable[n_] := Table[(-1)^IntegerDigits[i, 2, n], {i, 0, 2^n - 1}];
```

... and for the 3-period problem, the realizations are:

```
q = qTable[3];
TableForm[q, TableHeadings → {Automatic, {"q0", "q1", "q2"}},
 TableAlignments → Right]
```

	q_0	q_1	q_2
1	1	1	1
2	1	1	-1
3	1	-1	1
4	1	-1	-1
5	-1	1	1
6	-1	1	-1
7	-1	-1	1
8	-1	-1	-1

The transition probabilities for each path are:

```
PrTrans[q_] := Table[If[q[[i, j]] == q[[i, j - 1]], α, 1 - α],
   {i, Length[q]}, {j, 2, Dimensions[q][[2]]}];
qp = PrTrans[q];
TableForm[qp, TableHeadings → {Automatic, {"Pr0→1", "Pr1→2"}},
 TableAlignments → Right]
```

	$Pr_{0\to1}$	$Pr_{1\to2}$
1	α	α
2	α	$1-\alpha$
3	$1-\alpha$	$1-\alpha$
4	$1-\alpha$	α
5	$1-\alpha$	α
6	$1-\alpha$	$1-\alpha$
7	α	$1-\alpha$
8	α	α

The total probabilities of each path are:

```
TotalProbs = Apply[Times, qp, {1}] / 2;
TableForm[TotalProbs,
 TableHeadings → {Automatic, {"PrTotal"}}, TableAlignments → Right]
```

1	$\frac{\alpha^2}{2}$
2	$\frac{1}{2}(1-\alpha)\alpha$
3	$\frac{1}{2}(1-\alpha)^2$
4	$\frac{1}{2}(1-\alpha)\alpha$
5	$\frac{1}{2}(1-\alpha)\alpha$
6	$\frac{1}{2}(1-\alpha)^2$
7	$\frac{1}{2}(1-\alpha)\alpha$
8	$\frac{\alpha^2}{2}$

Verify that the probabilities sum to one:

```
Total[TotalProbs] // Simplify
```

1

Compute the v_t's using the definition $v_t = q_t - \phi q_{t-1}$:

```
v = q[[All, {2, 3}]] - φ q[[All, {1, 2}]];
TableForm[v, TableHeadings → {Automatic, {"v1", "v2"}}, TableAlignments → Right]
```

	v_1	v_2
1	$1-\phi$	$1-\phi$
2	$1-\phi$	$-1-\phi$
3	$-1-\phi$	$1+\phi$
4	$-1-\phi$	$-1+\phi$
5	$1+\phi$	$1-\phi$
6	$1+\phi$	$-1-\phi$
7	$-1+\phi$	$1+\phi$
8	$-1+\phi$	$-1+\phi$

Verify that $Ev_1 = Ev_2 = 0$:

```
TotalProbs.v[[All]] // Simplify
```

{0, 0}

$Var(v_t) \equiv \gamma_0$

```
γ0 = (TotalProbs.(v[[All]]^2) // Simplify)
```

$\{1 + (2 - 4\alpha)\phi + \phi^2, 1 + (2 - 4\alpha)\phi + \phi^2\}$

$Cov(v_t, v_{t-1}) = Ev_t v_{t-1} \equiv \gamma_1$

```
γ1 = (TotalProbs.(v[[All, 1]] v[[All, 2]]) // Simplify)
```

$-4\alpha^2\phi - (1+\phi)^2 + 2\alpha(1+\phi)^2$

The v_t must be uncorrelated, so solve for the value of ϕ that makes $\gamma_1 = 0$:

```
s = Solve[γ1 == 0, φ]
```

$$\left\{\left\{\phi \to \frac{1}{-1+2\alpha}\right\}, \{\phi \to -1+2\alpha\}\right\}$$

There are two solutions, but only the second has $|\phi| < 1$. For example,

```
φ /. s /. α → .6
```

{5., 0.2}

So, take the second solution:

```
s = s[[2, 1]]
```

$\phi \to -1 + 2\alpha$

Verify that this results in $\gamma_1 = 0$:

```
TotalProbs.(v[[All, 1]] v[[All, 2]]) /. s // Simplify
```

```
0
```

The v_t's have no skewness:

```
TotalProbs.(v[[All, 2]]^3) /. s // Simplify
```

```
0
```

Verify that the higher-order serial moment is non-zero, i.e., $\text{Cov}\left(v_{t-1}, v_t^3\right) = E v_{t-1} v_t^3 \neq 0$

```
TotalProbs.(v[[All, 1]] * v[[All, 2]]^3) /. s // Simplify
```

$$-32\,(-1+\alpha)^2\,\alpha^2\,(-1+2\,\alpha)$$

```
ClearAll[q, u, v, γ0, γ1]
```

Exercise 4.2 (Source notebook: RollBasicAndGeneralized)

The model is:

```
mRule = m_t_ :> m_(t-1) + u_t;
pRule = p_t_ :> m_t + c q_t;
ΔpRule = Δp_t_ :> (p_t /. pRule /. mRule) - (p_(t-1) /. pRule);
Δp_t /. ΔpRule
```

$$-c\,q_{-1+t} + c\,q_t + u_t$$

We need some alternate rules for the expectation operator to recognize the correlation between q_t and q_{t-1}.

```
ℰAlternateRules = {
   ℰ[q_^2] → 1, ℰ[u_^2] → σu^2,
   ℰ[q_ u_] → 0, ℰ[q_t_ q_s_] :> 0 /; Abs[t - s] > 1,
   ℰ[q_t_ q_s_] :> ρ /; Abs[t - s] == 1,
   ℰ[u_t_ u_s_] :> 0 /; t =!= s};
```

```
ℰRules = Join[ℰAlternateRules, ℰLinearityRules];
```

To get the variance, we multiply everything out, and take the expectation:

```
ℰ[Expand[Δp_t^2 /. ΔpRule]]
```

$$\mathcal{E}\left[c^2\,q_{-1+t}^2 - 2\,c^2\,q_{-1+t}\,q_t + c^2\,q_t^2 - 2\,c\,q_{-1+t}\,u_t + 2\,c\,q_t\,u_t + u_t^2\right]$$

Using the rules described about to eliminate terms that have zero expectation:

```
% //. ℰRules // Simplify
```

$-2\,c^2\,(-1+\rho)+\sigma_u^2$

Cov $(\Delta p_t, \Delta p_{t-1})$:

```
ℰ[Expand[Δp_t Δp_{t-1} /. ΔpRule]]
```

$\mathcal{E}[c^2\,q_{-2+t}\,q_{-1+t} - c^2\,q_{-1+t}^2 - c^2\,q_{-2+t}\,q_t + c^2\,q_{-1+t}\,q_t - c\,q_{-1+t}\,u_{-1+t} + c\,q_t\,u_{-1+t} - c\,q_{-2+t}\,u_t + c\,q_{-1+t}\,u_t + u_{-1+t}\,u_t]$

```
ℰ[Expand[Δp_t Δp_{t-1} /. ΔpRule]] //. ℰRules // FullSimplify
```

$c^2\,(-1+2\,\rho)$

```
ℰ[Expand[Δp_t Δp_{t-2} /. ΔpRule]] //. ℰRules // Simplify
```

$-c^2\,\rho$

```
ℰ[Expand[Δp_t Δp_{t-3} /. ΔpRule]] //. ℰRules // Simplify
```

0

Verify that $\sqrt{-\text{Cov}\,(\Delta p_t, \Delta p_{t-1})} < c$:

```
Simplify[Sqrt[-ℰ[Expand[Δp_t Δp_{t-1} /. ΔpRule]] //. ℰRules] < c,
 Assumptions → {0 < ρ < 1 / 2, c > 0}]
```

True

Exercise 4.3 (Source notebook: RollBasicAndGeneralized)

Model:

```
mRule = m_{t_} :→ m_{t-1} + u_t;
pRule = p_{t_} :→ m_t + c q_t;
ΔpRule = Δp_{t_} :→ (p_t /. pRule /. mRule) - (p_{t-1} /. pRule);
Δp_t /. ΔpRule
```

$-c\,q_{-1+t} + c\,q_t + u_t$

Modified expectations rules:

```
εAlternate_Rules = {
   ε[q_^2] → 1,
   ε[u_^2] → σu^2,
   ε[q_s_ u_t_] :→ ρ σu /; t == s,
   ε[q_s_ u_t_] :→ 0 /; t =!= s,
   ε[q_t_ q_s_] :→ 0 /; t =!= s,
   ε[u_t_ u_s_] :→ 0 /; t =!= s};
```

```
ε_Rules = Join[εAlternate_Rules, εLinearity_Rules];
```

```
ε[Expand[Δp_t^2 /. Δp_Rule]] //. ε_Rules // FullSimplify
```

$2\,c^2 + \sigma_u^2 + 2\,c\,\rho\,\sigma_u$

```
ε[Expand[Δp_t Δp_{t-1} /. Δp_Rule]] //. ε_Rules // Simplify
```

$-c\,(c + \rho\,\sigma_u)$

```
ε[Expand[Δp_t Δp_{t-2} /. Δp_Rule]] //. ε_Rules // Simplify
```

0

Verify that $\sqrt{-\mathrm{Cov}\,(\Delta p_t, \Delta p_{t-1})} > c$:

```
Simplify[Sqrt[-ε[Expand[Δp_t Δp_{t-1} /. Δp_Rule]] //. ε_Rules] > c,
 Assumptions → {0 < ρ, σu > 0, c > 0}]
```

True

Exercise 5.1 (Different μs for different brokers, Source notebook: SequentialTrade)

```
Ask
```

$$\frac{\mathrm{VLo}\,\delta\,(-1+\mu) + \mathrm{VHi}\,(-1+\delta)\,(1+\mu)}{-1+(-1+2\,\delta)\,\mu}$$

```
Ask /. δ → 1/2 // FullSimplify
```

$$\frac{1}{2}\,(\mathrm{VHi} + \mathrm{VLo} + \mathrm{VHi}\,\mu - \mathrm{VLo}\,\mu)$$

```
Profits_b = (Ask /. μ → μ_Other) - (Ask /. μ → μ_b) /. δ → 1 / 2 // FullSimplify
```

$$-\frac{1}{2}\,(\text{VHi} - \text{VLo})\,(\mu_b - \mu_{\text{Other}})$$

```
Simplify[Profits_b > 0, {0 < μ_b < 1, 0 < μ_Other < 1, μ_b < μ_Other, VHi > VLo, 0 < δ < 1}]
```

```
True
```

Exercise 5.2 (Informed trading only in low state, Source notebook: SequentialTrade)

```
LabelTree = {V, {VLo, {In, S}, {Un, B, S}}, {VHi, {Un, B, S}}};
LabelTree // ShowTree
```

```
V VLo In S
      Un B
         S
  VHi Un B
         S
```

```
PrTree = LabelTree /. {VLo, {In, S}, a_} :→ {VLo, {In, 1}, a} /. B | S → 1 / 2 /. In → μ /.
      {VHi, {Un, a__}} :→ {VHi, {1, a}} /. Un → 1 - μ /. VLo → δ /. VHi → 1 - δ /. V → 1;
ExTree = BuildTree[LabelTree, PrTree];
ShowTree[ExTree]
```

V 1 1	VLo δ δ	In μ $\delta\,\mu$	S 1 $\delta\,\mu$
		Un $1-\mu$ $\delta\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$
			S $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$
	VHi $1-\delta$ $1-\delta$	Un 1 $1-\delta$	B $\frac{1}{2}$ $\frac{1-\delta}{2}$
			S $\frac{1}{2}$ $\frac{1-\delta}{2}$

Verify that terminal node total probabilities add up to one:

```
Pr[ExTree, B] + Pr[ExTree, S] // Simplify
```

1

Conditional on a buy, the revised δ is:

```
δB = Pr[ExTree, VLo, B] / Pr[ExTree, B] // Simplify
```

$$\frac{\delta\,(-1+\mu)}{-1+\delta\,\mu}$$

Conditional on a sell ...

```
δS = Pr[ExTree, VLo, S] / Pr[ExTree, S] // Simplify
```

$$\frac{\delta\,(1+\mu)}{1+\delta\,\mu}$$

```
Ask = Simplify[δB VLo + (1 - δB) VHi]
```

$$\frac{\text{VHi}\,(-1+\delta)+\text{VLo}\,\delta\,(-1+\mu)}{-1+\delta\,\mu}$$

```
Bid = Simplify[δS VLo + (1 - δS) VHi]
```

$$\frac{\text{VHi}-\text{VHi}\,\delta+\text{VLo}\,\delta\,(1+\mu)}{1+\delta\,\mu}$$

```
Simplify[δS /. δ → δB]
```

$$\frac{\delta\,(-1+\mu)\,(1+\mu)}{-1+\delta\,\mu^2}$$

Manipulation by buying at the ask and selling at the revised bid? Nope.

```
Simplify[(-Ask + Bid /. δ → δB) > 0, {0 < μ < 1, 0 < δ < 1, VHi > VLo}]
```

False

Manipulation by selling at the bid and buying at the revised ask? Nope.

```
Simplify[(Bid - Ask /. δ → δS) > 0, {0 < μ < 1, 0 < δ < 1, VHi > VLo}]
```

False

Exercise 5.3 (Informed traders get a signal, Source notebook: SequentialTrade)

```
LabelTree =
 {V,
  {VLo,
   {In, {SLo, S}, {SHi, B}},
   {Un, B, S}},
  {VHi,
   {In, {SLo, S}, {SHi, B}},
   {Un, B, S}}
 }; LabelTree // ShowTree
```

```
V VLo In SLo S
         SHi B
      Un B
         S
  VHi In SLo S
         SHi B
      Un B
         S
```

```
PrTree =
  LabelTree /. {VLo, {In, {SLo, S}, {SHi, B}}, a_} :> {VLo, {In, {γ, 1}, {1 - γ, 1}},
           a} /. {VHi, {In, {SLo, S}, {SHi, B}}, a_} :>
          {VHi, {In, {1 - γ, 1}, {γ, 1}}, a} /. B | S → 1 / 2 /.
      In → μ /. Un → 1 - μ /. VLo → δ /. VHi → 1 - δ /. V → 1;
ExTree = BuildTree[LabelTree, PrTree];
ShowTree[ExTree]
```

V 1 1	VLo δ δ	In μ $\delta\,\mu$	SLo γ $\gamma\,\delta\,\mu$	S 1 $\gamma\,\delta\,\mu$
			SHi $1-\gamma$ $(1-\gamma)\,\delta\,\mu$	B 1 $(1-\gamma)\,\delta\,\mu$
		Un $1-\mu$ $\delta\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$	
			S $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$	
	VHi $1-\delta$ $1-\delta$	In μ $(1-\delta)\,\mu$	SLo $1-\gamma$ $(1-\gamma)\,(1-\delta)\,\mu$	S 1 $(1-\gamma)\,(1-\delta)\,\mu$
			SHi γ $\gamma\,(1-\delta)\,\mu$	B 1 $\gamma\,(1-\delta)\,\mu$
		Un $1-\mu$ $(1-\delta)\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,(1-\delta)\,(1-\mu)$	
			S $\frac{1}{2}$ $\frac{1}{2}\,(1-\delta)\,(1-\mu)$	

```
Pr[ExTree, B] + Pr[ExTree, S] // Simplify
```

1

```
δB = Pr[ExTree, VLo, B] / Pr[ExTree, B] // Simplify
```

$$\frac{\delta\,(-1+(-1+2\,\gamma)\,\mu)}{-1+(-1+2\,\gamma)\,(-1+2\,\delta)\,\mu}$$

```
Values = {γ → .1, μ → .2, δ → .3};
```

```
δB /. Values
```

0.371795

```
Ask = Simplify[δB VLo + (1 - δB) VHi]
```

$$\frac{\text{VLo}\,\delta\,(-1+(-1+2\gamma)\,\mu)+\text{VHi}\,(-1+\delta)\,(1+(-1+2\gamma)\,\mu)}{-1+(-1+2\gamma)\,(-1+2\delta)\,\mu}$$

```
δS = Pr[ExTree, VLo, S] / Pr[ExTree, S] // Simplify
```

$$\frac{\delta\,(1+(-1+2\gamma)\,\mu)}{1+(-1+2\gamma)\,(-1+2\delta)\,\mu}$$

```
Bid1 = Simplify[δS VLo + (1 - δS) VHi]
```

$$\frac{\text{VHi}\,(-1+\delta)\,(-1+(-1+2\gamma)\,\mu)+\text{VLo}\,\delta\,(1+(-1+2\gamma)\,\mu)}{1+(-1+2\gamma)\,(-1+2\delta)\,\mu}$$

```
Simplify[δS /. δ → δB]
```

δ

Exercise 5.4 (Offsetting trades, Source notebook: SequentialTrade)

A sell followed by a buy gives ... (see the analysis of the original problem, above).

```
FoldList[δCond, δ0, {S, B}] //. δCondRule // Simplify
```

$$\left\{\delta_0,\ \frac{(1+\mu)\,\delta_0}{1-\mu+2\,\mu\,\delta_0},\ \delta_0\right\}$$

Exercise 7.1 (Informative noise traders, Source notebook: StrategicTrade)

As in the basic problem:

```
pRule = p → y λ + μ;
yRule = y → u + x;
πRule = π → (v - p) x;
```

The informed trader's profits are:

```
π /. πRule /. pRule /. yRule
```

$x\,(v-(u+x)\,\lambda-\mu)$

At this point we diverge from the basic model because u and v are correlated. The projection the informed trader makes is:

```
uvDist = MVN[{{p0}, {0}}, {{Σ0, σuv}, {σuv, σu^2}}, {v, u}]
```

$$\begin{pmatrix} v \\ u \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ 0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & \sigma_{uv} \\ \sigma_{uv} & \sigma_u^2 \end{pmatrix}\right)$$

```
uConditionalDist = MVNConditional[uvDist, u, v]
```

$$u \sim \mathcal{N}\left(\frac{(v - p_0)\ \sigma_{uv}}{\Sigma_0}, \sigma_u^2 - \frac{\sigma_{uv}^2}{\Sigma_0}\right)$$

The informed trader's expected profits (conditional on v) are:

```
Eπ = π /. πRule /. pRule /. yRule /. u → GetMean[uConditionalDist]
```

$$x\left(v - \mu - \lambda\left(x + \frac{(v - p_0)\ \sigma_{uv}}{\Sigma_0}\right)\right)$$

The informed trader's optimal trade is:

```
xOpt = First @ Simplify[Solve[∂x Eπ == 0, x]]
```

$$\left\{x \to \frac{(v - \mu)\ \Sigma_0 + \lambda\ (-v + p_0)\ \sigma_{uv}}{2\ \lambda\ \Sigma_0}\right\}$$

The MM conjectures that the informed trader's demand is linear in v (as above), and must figure out E[v | y]:

```
xRule = x → α + v β;
xEqu = (x /. xRule) == (x /. xOpt)
```

$$\alpha + v\ \beta == \frac{(v - \mu)\ \Sigma_0 + \lambda\ (-v + p_0)\ \sigma_{uv}}{2\ \lambda\ \Sigma_0}$$

```
xSolutions = Reduce[ ∀v xEqu && Σ0 > 0 && λ > 0, {α, β}, Reals]
```

$$\Sigma_0 > 0\ \&\&\ \lambda > 0\ \&\&\ \alpha == \frac{-\mu\ \Sigma_0 + \lambda\ p_0\ \sigma_{uv}}{2\ \lambda\ \Sigma_0}\ \&\&\ \beta == \frac{\Sigma_0 - \lambda\ \sigma_{uv}}{2\ \lambda\ \Sigma_0}$$

```
xSolutions = Simplify @ ToRules @ Take[xSolutions, -2]
```

$$\left\{\alpha \to \frac{1}{2}\left(-\frac{\mu}{\lambda} + \frac{p_0\ \sigma_{uv}}{\Sigma_0}\right), \beta \to \frac{\Sigma_0 - \lambda\ \sigma_{uv}}{2\ \lambda\ \Sigma_0}\right\}$$

Now the MM must compute E[v | y]:

```
MakeLinearForm[uvDist, {v, y /. yRule /. xRule}]
```

$$\begin{pmatrix} v \\ u + \alpha + v\,\beta \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ \alpha + \beta\, p_0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & \beta\,\Sigma_0 + \sigma_{uv} \\ \beta\,\Sigma_0 + \sigma_{uv} & \sigma_u^2 + \beta\,\sigma_{uv} + \beta\,(\beta\,\Sigma_0 + \sigma_{uv}) \end{pmatrix}\right)$$

```
vyDist = SetLabel[%, {v, y}]
```

$$\begin{pmatrix} v \\ y \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ \alpha + \beta\, p_0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & \beta\,\Sigma_0 + \sigma_{uv} \\ \beta\,\Sigma_0 + \sigma_{uv} & \sigma_u^2 + \beta\,\sigma_{uv} + \beta\,(\beta\,\Sigma_0 + \sigma_{uv}) \end{pmatrix}\right)$$

```
vConditionalDist = MVNConditional[vyDist, v, y] // Simplify
```

$$v \sim \mathcal{N}\left(\frac{(y - \alpha)\,(\beta\,\Sigma_0 + \sigma_{uv}) + p_0\,(\sigma_u^2 + \beta\,\sigma_{uv})}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}, \frac{\Sigma_0\,\sigma_u^2 - \sigma_{uv}^2}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}\right)$$

Market efficiency:

```
pEqu = GetMean[vConditionalDist] == (p /. pRule)
```

$$\frac{(y - \alpha)\,(\beta\,\Sigma_0 + \sigma_{uv}) + p_0\,(\sigma_u^2 + \beta\,\sigma_{uv})}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}} == y\,\lambda + \mu$$

```
r = Reduce[ForAll[y, True, pEqu], {μ, λ}]
```

$$\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv} \neq 0\ \&\&\ \mu == \frac{-\alpha\,\beta\,\Sigma_0 + \sigma_u^2\,p_0 - \alpha\,\sigma_{uv} + \beta\,p_0\,\sigma_{uv}}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}\ \&\&\ \lambda == \frac{\beta\,\Sigma_0 + \sigma_{uv}}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}$$

```
pSolutions = ToRules[Take[r, -2]]
```

$$\left\{\mu \to \frac{-\alpha\,\beta\,\Sigma_0 + \sigma_u^2\,p_0 - \alpha\,\sigma_{uv} + \beta\,p_0\,\sigma_{uv}}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}, \lambda \to \frac{\beta\,\Sigma_0 + \sigma_{uv}}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}\right\}$$

Collecting the results and solving:

```
EquSet = Apply[Equal, Join[pSolutions, xSolutions], 1]
```

$$\left\{\mu == \frac{-\alpha\,\beta\,\Sigma_0 + \sigma_u^2\,p_0 - \alpha\,\sigma_{uv} + \beta\,p_0\,\sigma_{uv}}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}, \lambda == \frac{\beta\,\Sigma_0 + \sigma_{uv}}{\beta^2\,\Sigma_0 + \sigma_u^2 + 2\,\beta\,\sigma_{uv}}, \alpha == \frac{1}{2}\left(-\frac{\mu}{\lambda} + \frac{p_0\,\sigma_{uv}}{\Sigma_0}\right), \beta == \frac{\Sigma_0 - \lambda\,\sigma_{uv}}{2\,\lambda\,\Sigma_0}\right\}$$

```
Model_Solutions = Simplify[Solve[EquSet, {μ, λ, α, β}], {σu^2 > 0, Σ0 > 0}];
Model_Solutions // Transpose // TableForm
```

$\alpha \to \frac{p_0 \left(\sigma_{uv} - \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}\right)}{2 \Sigma_0}$	$\alpha \to \frac{p_0 \left(\sigma_{uv} + \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}\right)}{2 \Sigma_0}$
$\mu \to p_0$	$\mu \to p_0$
$\lambda \to \frac{\Sigma_0 \left(\sigma_{uv} + \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}\right)}{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2 + \sigma_{uv} \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}}$	$\lambda \to \frac{\Sigma_0 \left(-\sigma_{uv} + \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}\right)}{-4 \Sigma_0 \sigma_u^2 + 3 \sigma_{uv}^2 + \sigma_{uv} \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}}$
$\beta \to \frac{-\sigma_{uv} + \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}}{2 \Sigma_0}$	$\beta \to -\frac{\sigma_{uv} + \sqrt{4 \Sigma_0 \sigma_u^2 - 3 \sigma_{uv}^2}}{2 \Sigma_0}$

Only the first solution can have $\beta > 0$, So:

```
Model_Solutions = Model_Solutions[[1]];
```

When $\sigma_{uv} = 0$, this reduces to the original solutions:

```
Simplify[Model_Solutions /. σuv → 0, {Σ0 > 0, σu^2 > 0}]
```

$$\left\{\alpha \to -\frac{\sigma_u^2 p_0}{\sqrt{\Sigma_0 \sigma_u^2}},\ \mu \to p_0,\ \lambda \to \frac{\Sigma_0}{2 \sqrt{\Sigma_0 \sigma_u^2}},\ \beta \to \sqrt{\frac{\sigma_u^2}{\Sigma_0}}\right\}$$

With perfect correlation...

```
AltSolution = Simplify[Model_Solutions /. σuv → Sqrt[σu^2 Σ0], {Σ0 > 0, σu^2 > 0}]
```

$$\left\{\alpha \to 0,\ \mu \to p_0,\ \lambda \to \sqrt{\frac{\Sigma_0}{\sigma_u^2}},\ \beta \to 0\right\}$$

Since $\beta = 0$, the informed trader doesn't trade at all. The uninformed trade, though, is linear in v:

```
u_Rule = u → (v - p0) Sqrt[σu^2 Σ0] / Σ0;
```

Under these conditions, the market clearing price becomes $p = p_0 + \lambda y = v$.

```
Simplify[p_Rule /. AltSolution /. y_Rule /. x → 0 /. u_Rule, {Σ0 > 0, σu^2 > 0}]
```

$p \to v$

Also, the conditional variance is:

```
GetVariance[vConditionalDist] //. Model_Solutions /. σuv → Sqrt[σu^2 Σ0]
```

0

Exercise 7.2 (Informed trader gets a signal, Source notebook: StrategicTrade)

The informed trader in the basic model has perfect information about *v*. Consider the case where she only gets a signal *s* about *v*. That is, $s = v + \epsilon$ where $\epsilon \sim N[0, \sigma_\epsilon^2]$, independent of v. Solve the model by proceeding as in the basic case. Solve the informed trader's problem; solve the MM's problem; solve for the model parameters $(\alpha, \beta, \mu, \lambda)$ in terms of the inputs, σ_u^2, Σ_0, and σ_ϵ^2. Interpret your results. Verify that when $\sigma_\epsilon^2 = 0$, you get the original model solutions.

Solution

```
pRule = p → y λ + μ;
yRule = y → u + x;
πRule = π → (v - p) x;
```

Informed trader's profits:

```
π /. πRule /. pRule /. yRule
```

$x\,(v - (u + x)\,\lambda - \mu)$

The informed trader gets the signal s:

```
sRule = s → v + ϵ;
```

There are three random variables in this problem:

```
vϵuDistribution = MVN[{p0, 0, 0}, {{Σ0, 0, 0}, {0, σϵ^2, 0}, {0, 0, σu^2}}, {v, ϵ, u}]
```

$$\begin{pmatrix} v \\ \epsilon \\ u \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ 0 \\ 0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & 0 & 0 \\ 0 & \sigma_\epsilon^2 & 0 \\ 0 & 0 & \sigma_u^2 \end{pmatrix}\right)$$

We can rework this into a distribution for v, s, u:

```
MakeLinearForm[vϵuDistribution, {v, s /. sRule, u}]
```

$$\begin{pmatrix} v \\ v + \epsilon \\ u \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ p_0 \\ 0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & \Sigma_0 & 0 \\ \Sigma_0 & \Sigma_0 + \sigma_\epsilon^2 & 0 \\ 0 & 0 & \sigma_u^2 \end{pmatrix}\right)$$

```
vsuDistribution = SetLabel[%, {v, s, u}]
```

$$\begin{pmatrix} v \\ s \\ u \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ p_0 \\ 0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & \Sigma_0 & 0 \\ \Sigma_0 & \Sigma_0 + \sigma_\epsilon^2 & 0 \\ 0 & 0 & \sigma_u^2 \end{pmatrix}\right)$$

The informed trader forms the conditional distribution of v based on his signal:

```
vConditionalDistInf = MVNConditional[vsuDistribution, v, s]
```

$$v \sim \mathcal{N}\left(\frac{\Sigma_0\,(s - p_0)}{\Sigma_0 + \sigma_\epsilon^2} + p_0,\ \Sigma_0 - \frac{\Sigma_0^2}{\Sigma_0 + \sigma_\epsilon^2}\right)$$

The expected profits are developed from:

```
π /. πRule /. pRule /. yRule
```

$$x\,(v - (u + x)\,\lambda - \mu)$$

... substituting in the conditional mean for v:

```
Eπ = π /. πRule /. pRule /. yRule /. u → 0 /. v → GetMean[vConditionalDistInf]
```

$$x\left(-x\,\lambda - \mu + \frac{\Sigma_0\,(s - p_0)}{\Sigma_0 + \sigma_\epsilon^2} + p_0\right)$$

The informed trader maximizes expected profits by trading *x*:

```
xOpt = First @ Simplify[Solve[∂x Eπ == 0, x]]
```

$$\left\{x \to \frac{s\,\Sigma_0 - \mu\left(\Sigma_0 + \sigma_\epsilon^2\right) + \sigma_\epsilon^2\,p_0}{2\,\lambda\left(\Sigma_0 + \sigma_\epsilon^2\right)}\right\}$$

The MM conjectures that the informed trader's demand is linear in s:

```
xRule = x → α + s β;
```

Knowing the optimization process that the informed trader followed, the MM can solve for α and β:

```
xEqu = (x /. xRule) == (x /. xOpt)
```

$$\alpha + s\,\beta == \frac{s\,\Sigma_0 - \mu\left(\Sigma_0 + \sigma_\epsilon^2\right) + \sigma_\epsilon^2\,p_0}{2\,\lambda\left(\Sigma_0 + \sigma_\epsilon^2\right)}$$

```
xSolutions = Reduce[∀s xEqu && Σ0 > 0 && σϵ^2 > 0 && λ > 0 && s ≠ 0, {α, β}, Reals]
```

$$\sigma_\epsilon^2 > 0\ \&\&\ \Sigma_0 > 0\ \&\&\ \lambda > 0\ \&\&\ \left(\left(s < 0\ \&\&\ \alpha == \frac{-\mu\,\Sigma_0 - \mu\,\sigma_\epsilon^2 + \sigma_\epsilon^2\,p_0}{2\,\lambda\,\Sigma_0 + 2\,\lambda\,\sigma_\epsilon^2}\ \&\&\ \beta == \frac{\Sigma_0}{2\,\lambda\,\Sigma_0 + 2\,\lambda\,\sigma_\epsilon^2}\right)\ ||\ \left(s > 0\ \&\&\ \alpha == \frac{-\mu\,\Sigma_0 - \mu\,\sigma_\epsilon^2 + \sigma_\epsilon^2\,p_0}{2\,\lambda\,\Sigma_0 + 2\,\lambda\,\sigma_\epsilon^2}\ \&\&\ \beta == \frac{\Sigma_0}{2\,\lambda\,\Sigma_0 + 2\,\lambda\,\sigma_\epsilon^2}\right)\right)$$

```
xSolutions = Simplify @ ToRules @ Take[xSolutions[[4, 1]], -2]
```

$$\left\{\alpha \to \frac{-\mu\,(\Sigma_0 + \sigma_\epsilon^2) + \sigma_\epsilon^2\, p_0}{2\,\lambda\,(\Sigma_0 + \sigma_\epsilon^2)},\ \beta \to \frac{\Sigma_0}{2\,\lambda\,\Sigma_0 + 2\,\lambda\,\sigma_\epsilon^2}\right\}$$

Now the MM must figure out $E[V \mid Y = y]$. This is a little more involved than in the original problem because the informed trader's demand is conditioned on s. The joint distribution of v and $y = u + \alpha + s\beta$ is:

```
MakeLinearForm[vsuDistribution, {v, y /. yRule /. xRule}]
```

$$\begin{pmatrix} v \\ u + \alpha + s\,\beta \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ \alpha + \beta\, p_0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & \beta\,\Sigma_0 \\ \beta\,\Sigma_0 & \sigma_u^2 + \beta^2\,(\Sigma_0 + \sigma_\epsilon^2) \end{pmatrix}\right)$$

... and relabeling:

```
vyDistribution = SetLabel[%, {v, y}]
```

$$\begin{pmatrix} v \\ y \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ \alpha + \beta\, p_0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & \beta\,\Sigma_0 \\ \beta\,\Sigma_0 & \sigma_u^2 + \beta^2\,(\Sigma_0 + \sigma_\epsilon^2) \end{pmatrix}\right)$$

So the distribution of v (conditional on y) is:

```
vConditionalDistributionMM = MVNConditional[vyDistribution, v, y]
```

$$v \sim \mathcal{N}\left(p_0 + \frac{\beta\,\Sigma_0\,(y - \alpha - \beta\, p_0)}{\sigma_u^2 + \beta^2\,(\Sigma_0 + \sigma_\epsilon^2)},\ \Sigma_0 - \frac{\beta^2\,\Sigma_0^2}{\sigma_u^2 + \beta^2\,(\Sigma_0 + \sigma_\epsilon^2)}\right)$$

Market efficiency requires

```
pEqu = GetMean[vConditionalDistributionMM] == (p /. pRule)
```

$$p_0 + \frac{\beta\,\Sigma_0\,(y - \alpha - \beta\, p_0)}{\sigma_u^2 + \beta^2\,(\Sigma_0 + \sigma_\epsilon^2)} == y\,\lambda + \mu$$

Solving:

```
pSolutions = Reduce[∀y pEqu && Σ0 > 0 && σϵ² > 0 && σu² > 0, {μ, λ}, Reals]
```

$$\sigma_\epsilon^2 > 0 \,\&\&\, \Sigma_0 > 0 \,\&\&\, \sigma_u^2 > 0 \,\&\&\, \mu == \frac{-\alpha\,\beta\,\Sigma_0 + \sigma_u^2\, p_0 + \beta^2\,\sigma_\epsilon^2\, p_0}{\beta^2\,\Sigma_0 + \sigma_u^2 + \beta^2\,\sigma_\epsilon^2} \,\&\&\, \lambda == \frac{\beta\,\Sigma_0}{\beta^2\,\Sigma_0 + \sigma_u^2 + \beta^2\,\sigma_\epsilon^2}$$

```
pSolutions = Simplify @ ToRules @ Take[pSolutions, -2]
```

$$\left\{\mu \to \frac{-\alpha\,\beta\,\Sigma_0 + (\sigma_u^2 + \beta^2\,\sigma_\epsilon^2)\, p_0}{\sigma_u^2 + \beta^2\,(\Sigma_0 + \sigma_\epsilon^2)},\ \lambda \to \frac{\beta\,\Sigma_0}{\sigma_u^2 + \beta^2\,(\Sigma_0 + \sigma_\epsilon^2)}\right\}$$

Collecting the results and solving:

```
EquationSet = Apply[Equal, Join[pSolutions, xSolutions], 1]
```

$$\left\{\mu == \frac{-\alpha\,\beta\,\Sigma_0 + \left(\sigma_u^2 + \beta^2\,\sigma_\epsilon^2\right) p_0}{\sigma_u^2 + \beta^2\,\left(\Sigma_0 + \sigma_\epsilon^2\right)},\ \lambda == \frac{\beta\,\Sigma_0}{\sigma_u^2 + \beta^2\,\left(\Sigma_0 + \sigma_\epsilon^2\right)},\ \alpha == \frac{-\mu\,\left(\Sigma_0 + \sigma_\epsilon^2\right) + \sigma_\epsilon^2\,p_0}{2\,\lambda\,\left(\Sigma_0 + \sigma_\epsilon^2\right)},\ \beta == \frac{\Sigma_0}{2\,\lambda\,\Sigma_0 + 2\,\lambda\,\sigma_\epsilon^2}\right\}$$

```
ModelSolutions =
 Simplify[Solve[EquationSet, {μ, λ, α, β}], {Σ0 > 0, σu^2 > 0, σϵ^2 > 0}] // First
```

$$\left\{\alpha \to -\sqrt{\frac{\sigma_u^2}{\Sigma_0 + \sigma_\epsilon^2}}\ p_0,\ \mu \to p_0,\ \lambda \to \frac{\Sigma_0}{2\,\sqrt{\sigma_u^2\,\left(\Sigma_0 + \sigma_\epsilon^2\right)}},\ \beta \to \sqrt{\frac{\sigma_u^2}{\Sigma_0 + \sigma_\epsilon^2}}\right\}$$

To recover the original solutions:

```
ModelSolutions /. σϵ^2 → 0
```

$$\left\{\alpha \to -\sqrt{\frac{\sigma_u^2}{\Sigma_0}}\ p_0,\ \mu \to p_0,\ \lambda \to \frac{\Sigma_0}{2\,\sqrt{\Sigma_0\,\sigma_u^2}},\ \beta \to \sqrt{\frac{\sigma_u^2}{\Sigma_0}}\right\}$$

Given the price (or equivalently the total order flow), the variance of v is:

```
GetVariance[vConditionalDistributionMM]
```

$$\Sigma_0 - \frac{\beta^2\,\Sigma_0^2}{\sigma_u^2 + \beta^2\,\left(\Sigma_0 + \sigma_\epsilon^2\right)}$$

As we degrade the signal (increase σ_ϵ^2), the conditional variance approaches the unconditional variance.

Exercise 7.3 (Broker piggy-backs on informed trader, Source notebook: StrategicTrade)

Solution

```
pRule = p → y λ + μ;
```

But now the order flow includes the broker's order flow γx.

```
yRule = y → u + x (1 + γ);
```

The informed trader's profits are:

```
πRule = π → (v - p) x;
```

Substituting in for the price conjecture and *y*:

```
π /. π_Rule /. p_Rule /. y_Rule
```

$x\,(v - (u + x\,(1 + \gamma))\,\lambda - \mu)$

The expected profits (conditional on *v*) are $E\pi$:

```
Eπ = Simplify[π /. π_Rule /. p_Rule /. y_Rule /. u → 0]
```

$x\,(v - x\,(1 + \gamma)\,\lambda - \mu)$

The optimal quantity is:

```
xOpt = First @ Solve[∂_x Eπ == 0, x]
```

$\left\{x \to \frac{v - \mu}{2\,(1 + \gamma)\,\lambda}\right\}$

```
x_Rule = x → α + v β;
xEqu = (x /. x_Rule) == (x /. xOpt)
```

$\alpha + v\,\beta == \frac{v - \mu}{2\,(1 + \gamma)\,\lambda}$

Solving:

```
x_Solutions = Reduce[ ∀_v xEqu &&  λ > 0 && γ > 0 , {α, β}, Reals]
```

$\lambda > 0\ \&\&\ \gamma > 0\ \&\&\ \alpha == -\frac{\mu}{2\,\lambda + 2\,\gamma\,\lambda}\ \&\&\ \beta == \frac{1}{2\,\lambda + 2\,\gamma\,\lambda}$

```
x_Solutions = Simplify @ ToRules @ Take[x_Solutions , -2]
```

$\left\{\alpha \to -\frac{\mu}{2\,\lambda + 2\,\gamma\,\lambda},\ \beta \to \frac{1}{2\,\lambda + 2\,\gamma\,\lambda}\right\}$

As always, the MM must figure out $E[v \mid y]$. Starting from the original joint distribution:

```
uvDist = MVN[{{p_0}, {0}}, {{Σ_0, 0}, {0, σ_u^2}}, {v, u}]
```

$\begin{pmatrix} v \\ u \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ 0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & 0 \\ 0 & \sigma_u^2 \end{pmatrix}\right)$

Now, though, we have a more complicated form for $y : y = u + (\alpha + v\,\beta)\,(1 + \gamma)$. The joint distribution of v and y is:

```
MakeLinearForm[uvDist, {v, y /. yRule /. xRule}]
```

$$\begin{pmatrix} v \\ u + (\alpha + v\beta)(1+\gamma) \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ (\alpha + v\beta)(1+\gamma) - v(\beta+\beta\gamma) + (\beta+\beta\gamma)\,p_0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & (\beta+\beta\gamma)\,\Sigma_0 \\ (\beta+\beta\gamma)\,\Sigma_0 & (\beta+\beta\gamma)^2\,\Sigma_0 + \sigma_u^2 \end{pmatrix}\right)$$

...relabeling:

```
vyDistribution = SetLabel[%, {v, y}]
```

$$\begin{pmatrix} v \\ y \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} p_0 \\ (\alpha + v\beta)(1+\gamma) - v(\beta+\beta\gamma) + (\beta+\beta\gamma)\,p_0 \end{pmatrix}, \begin{pmatrix} \Sigma_0 & (\beta+\beta\gamma)\,\Sigma_0 \\ (\beta+\beta\gamma)\,\Sigma_0 & (\beta+\beta\gamma)^2\,\Sigma_0 + \sigma_u^2 \end{pmatrix}\right)$$

The conditional distribution is:

```
vConditionalDist = MVNConditional[vyDistribution, v, y] // Simplify
```

$$v \sim \mathcal{N}\left(\frac{-\beta(1+\gamma)(-y+\alpha+\alpha\gamma)\,\Sigma_0 + \sigma_u^2\,p_0}{\beta^2(1+\gamma)^2\,\Sigma_0 + \sigma_u^2}, \frac{\Sigma_0\,\sigma_u^2}{\beta^2(1+\gamma)^2\,\Sigma_0 + \sigma_u^2}\right)$$

Market efficiency:

```
pEqu = GetMean[vConditionalDist] == (p /. pRule)
```

$$\frac{-\beta(1+\gamma)(-y+\alpha+\alpha\gamma)\,\Sigma_0 + \sigma_u^2\,p_0}{\beta^2(1+\gamma)^2\,\Sigma_0 + \sigma_u^2} == y\,\lambda + \mu$$

```
pSolutions = Reduce[∀y pEqu, {μ, λ}]
```

$$\beta^2\,\Sigma_0 + 2\,\beta^2\,\gamma\,\Sigma_0 + \beta^2\,\gamma^2\,\Sigma_0 + \sigma_u^2 \neq 0 \;\&\&\; \mu == \frac{-\alpha\,\beta\,\Sigma_0 - 2\,\alpha\,\beta\,\gamma\,\Sigma_0 - \alpha\,\beta\,\gamma^2\,\Sigma_0 + \sigma_u^2\,p_0}{\beta^2\,\Sigma_0 + 2\,\beta^2\,\gamma\,\Sigma_0 + \beta^2\,\gamma^2\,\Sigma_0 + \sigma_u^2} \;\&\&\; \lambda == \frac{\beta\,\Sigma_0 + \beta\,\gamma\,\Sigma_0}{\beta^2\,\Sigma_0 + 2\,\beta^2\,\gamma\,\Sigma_0 + \beta^2\,\gamma^2\,\Sigma_0 + \sigma_u^2}$$

```
pSolutions = Simplify @ ToRules[Take[pSolutions, -2]]
```

$$\left\{\mu \to \frac{-\alpha\,\beta(1+\gamma)^2\,\Sigma_0 + \sigma_u^2\,p_0}{\beta^2(1+\gamma)^2\,\Sigma_0 + \sigma_u^2}, \lambda \to \frac{\beta(1+\gamma)\,\Sigma_0}{\beta^2(1+\gamma)^2\,\Sigma_0 + \sigma_u^2}\right\}$$

```
EquationSet = Apply[Equal, Join[pSolutions, xSolutions], 1]
```

$$\left\{\mu == \frac{-\alpha\,\beta(1+\gamma)^2\,\Sigma_0 + \sigma_u^2\,p_0}{\beta^2(1+\gamma)^2\,\Sigma_0 + \sigma_u^2}, \lambda == \frac{\beta(1+\gamma)\,\Sigma_0}{\beta^2(1+\gamma)^2\,\Sigma_0 + \sigma_u^2}, \alpha == -\frac{\mu}{2\,\lambda + 2\,\gamma\,\lambda}, \beta == \frac{1}{2\,\lambda + 2\,\gamma\,\lambda}\right\}$$

```
Model_Solutions = Simplify[Solve[EquationSet, {μ, λ, α, β}], {Σ0 > 0, σu^2 > 0}] // First
```

$$\left\{\alpha \to -\frac{\sqrt{\frac{\sigma_u^2}{\Sigma_0}}\ p_0}{1+\gamma},\ \mu \to p_0,\ \lambda \to \frac{1}{2}\sqrt{\frac{\Sigma_0}{\sigma_u^2}},\ \beta \to \frac{\sqrt{\frac{\sigma_u^2}{\Sigma_0}}}{1+\gamma}\right\}$$

To recover the original solution ...

```
Model_Solutions /. γ → 0
```

$$\left\{\alpha \to -\sqrt{\frac{\sigma_u^2}{\Sigma_0}}\ p_0,\ \mu \to p_0,\ \lambda \to \frac{1}{2}\sqrt{\frac{\Sigma_0}{\sigma_u^2}},\ \beta \to \sqrt{\frac{\sigma_u^2}{\Sigma_0}}\right\}$$

The expected profits are

```
Simplify[PowerExpand[Eπ /. x_Rule /. Model_Solutions]]
```

$$\frac{\sqrt{\sigma_u^2}\ (v-p_0)^2}{2\ (1+\gamma)\ \sqrt{\Sigma_0}}$$

The informed trader's demand is:

```
Simplify[x_Rule /. Model_Solutions]
```

$$x \to \frac{\sqrt{\frac{\sigma_u^2}{\Sigma_0}}\ (v-p_0)}{1+\gamma}$$

Since $\gamma > 0$, the informed trader's expected profits are lower. Also, she fades her demand to take into account the crooked broker. The informativeness of the price is:

```
GetVariance[vConditionalDist] /. Model_Solutions
```

$$\frac{\Sigma_0}{2}$$

Unchanged, relative to the original model.

Exercise 8.1 (Source notebook: RollBasicAndGeneralized)

The model is observationally equivalent to one in which there is no lag on the efficient price. The autocovariances and moving average representation are the same.

Exercise 8.2 (Source notebook: RollBasicAndGeneralized)

By rearranging, the model can be written as $(1-(1-\alpha)L)p_t = \alpha m_t$. Taking first differences $(1-(1-\alpha)L)\triangle p_t = \epsilon_t = \alpha w_t$ So:

```
ϕ[L_] := 1 - (1 - α) L
```

The MA representation is $\triangle p_t = \Theta(L)\epsilon_t$ where $\Theta(L) = \phi(L)^{-1}$. Furthermore $\Theta(1)^2 =$

```
ϕ[1]^-2
```

$$\frac{1}{\alpha^2}$$

Since $\sigma_\epsilon^2 = \alpha^2 \sigma_w^2$, $\Theta(1)^2 \sigma_\epsilon^2 = \sigma_w^2$

Exercise 8.3 (Source notebook: RollBasicAndGeneralized)

Over five-minute intervals

```
σRule = σϵ^2 → 0.00001;
```

```
θ[L_] := 1 - 0.3 L + 0.1 L^2
```

Random-walk variance:

```
Sqrt[θ[1]^2 σϵ^2 /. σRule]
```

```
0.00252982
```

Over one day:

```
Sqrt[6 * 12 * θ[1]^2 σϵ^2 /. σRule]
```

```
0.0214663
```

i.e., about 2%

For the pricing error variance, the C_i coefficients are generally:

```
CRule = C[θ_, i_] :> Sum[-Coefficient[θ[L], L, j], {j, i+1, Exponent[θ[L], L]}];
```

and here ...

```
Table[C[θ, i], {i, 0, 2}] /. CRule
```

```
{0.2, -0.1, 0}
```

$$\sqrt{\sum_{i=0}^{\text{Exponent}[\theta[L],L]-1} C[\theta, i]^2 \, \sigma_\epsilon^2 \;/.\; C_{Rule} \;/.\; \sigma_{Rule}}$$

```
0.000707107
```

i.e., about seven basis points

Exercise 8.4 (Source notebook: RollBasicAndGeneralized)

The structural model is:

$m_t = m_{t-1} + w_t$

$w_t = \lambda q_t + u_t$

$p_t = m_{t-1} + c\, q_t$

Notice that the price is determined with respect to *lagged* value of the implicit efficient price.

(a) Using the structural representation, determine the Δp_t autocovariances γ_0, γ_1 and verify that $\gamma_2 = 0$.

In this and the following parts, assume that $c = 2$ and $\lambda = 1$.

(b) Verify that autocovariances are the same as the autocovariances for the (statistical) MA(1) model $\Delta p_t = \epsilon_t + \Theta$ where

$$\sigma_\epsilon^2 = \frac{1}{2}\left(\sigma_u^2 + \sqrt{\left(\sigma_u^2+1\right)\left(\sigma_u^2+9\right)} + 5\right)$$

and

$$\theta = \frac{1}{4}\left(-\sigma_u^2 + \sqrt{\left(\sigma_u^2+1\right)\left(\sigma_u^2+9\right)} - 5\right)$$

(c) Verify that $\sigma_w^2 = (1+\Theta)^2 \sigma_\epsilon^2$.

(d) Compute (in terms of the MA parameters) the lower bound for σ_s^2 where $s_t = p_t - m_t$. Verify that the lower bound is exact when $\sigma_u^2 = 0$.

Analysis

```
nValues = {c → 2, λ → 1};
```

```
mRule = m_t_ :→ m_(t-1) + w_t;
wRule = w_t_ :→ λ q_t + u_t;
pRule = p_t_ :→ m_(t-1) + c q_t;
```

Pricing error $s_t =$

```
pt - mt /. pRule /. mt → (mt /. mRule) /. wRule
```

$c\, q_t - \lambda\, q_t - u_t$

This implies that the pricing error variance is

```
(c - λ)^2 + σu^2 /. nValues
```

$1 + \sigma_u^2$

Price changes:

```
ΔpRule = Δpt_ :→ (pt /. pRule /. mRule /. wRule) - (pt-1 /. pRule);
Δpt /. ΔpRule
```

$-c\, q_{-1+t} + \lambda\, q_{-1+t} + c\, q_t + u_{-1+t}$

$\mathrm{Var}[\Delta p_t] = \gamma_0$:

```
ℰ[Expand[Δpt^2 /. ΔpRule]] //. ℰRules
```

$2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2$

```
2 c^2 - 2 λ c + λ^2 + σu^2
```

$2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2$

$\mathrm{Cov}[\Delta p_t, \Delta p_{t-1}] = \gamma_1$:

```
ℰ[Expand[Δpt Δpt-1 /. ΔpRule]] //. ℰRules
```

$-c^2 + c\, \lambda$

$\mathrm{Cov}[\Delta p_t, \Delta p_{t-2}] = \gamma_2$:

```
ℰ[Expand[Δpt Δpt-2 /. ΔpRule]] //. ℰRules
```

0

Summarize the first two autocovariances in terms of the structural parameters:

```
γStructuralRules = {γ0 → 2 c^2 - 2 λ c + λ^2 + σu^2, γ1 → -c^2 + c λ};
γStructuralRules // TableForm
```

$\gamma_0 \to 2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2$
$\gamma_1 \to -c^2 + c\, \lambda$

Now evaluate $\mathrm{Var}[w_t] = \sigma_w^2$:

```
ℰ[Expand[w_t^2 /. w_Rule]] //. ℰ_Rules
```

$\lambda^2 + \sigma_u^2$

The autocovariances computed from the statistical and structural representations must agree. The autocovariances for the MA(1) process $\triangle p_t = \epsilon_t + \theta\epsilon_{t-1}$ are:

```
γStatistical_Rules = {γ_0 → (θ^2 + 1) σ_ε^2, γ_1 → θ σ_ε^2}; γStatistical_Rules // TableForm
```

$\gamma_0 \to (1 + \theta^2)\, \sigma_\epsilon^2$

$\gamma_1 \to \theta\, \sigma_\epsilon^2$

```
StatStructEqu = Apply[Equal, Join[γStatistical_Rules, γStructural_Rules], {1}];
StatStructEqu // TableForm
```

$\gamma_0 == (1 + \theta^2)\, \sigma_\epsilon^2$

$\gamma_1 == \theta\, \sigma_\epsilon^2$

$\gamma_0 == 2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2$

$\gamma_1 == -c^2 + c\, \lambda$

which implies:

```
sol = Solve[StatStructEqu, {θ, σ_ε^2}, {γ_0, γ_1}] // Simplify
```

$$\left\{\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2 + \sigma_u^2}\, \sqrt{4\, c^2 - 4\, c\, \lambda + \lambda^2 + \sigma_u^2}\right), \theta \to -\frac{2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2 + \sigma_u^2}\, \sqrt{4\, c^2 - 4\, c\, \lambda + \lambda^2 + \sigma_u^2}}{2\, c^2 - 2\, c\, \lambda}\right\}, \left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2 + \sigma_u^2}\, \sqrt{4\, c^2 - 4\, c\, \lambda + \lambda^2 + \sigma_u^2}\right), \theta \to -\frac{2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2 + \sigma_u^2}\, \sqrt{4\, c^2 - 4\, c\, \lambda + \lambda^2 + \sigma_u^2}}{2\, c^2 - 2\, c\, \lambda}\right\}\right\}$$

```
InvertibleSolution = sol[[2]]
```

$$\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2 + \sigma_u^2}\, \sqrt{4\, c^2 - 4\, c\, \lambda + \lambda^2 + \sigma_u^2}\right), \theta \to -\frac{2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2 + \sigma_u^2}\, \sqrt{4\, c^2 - 4\, c\, \lambda + \lambda^2 + \sigma_u^2}}{2\, c^2 - 2\, c\, \lambda}\right\}$$

```
FullSimplify[InvertibleSolution /. nValues, {λ > 0, σ_u^2 > 0, c > 0}]
```

$$\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(5 + \sigma_u^2 + \sqrt{(1 + \sigma_u^2)\,(9 + \sigma_u^2)}\right), \theta \to \frac{1}{4}\left(-5 - \sigma_u^2 + \sqrt{(1 + \sigma_u^2)\,(9 + \sigma_u^2)}\right)\right\}$$

```
Simplify[ (1 + θ)^2 σϵ^2 /. InvertibleSolution]
```

$\lambda^2 + \sigma_u^2$

i.e., the coefficient of ϵ_t is the same in both representations.

```
Simplify[θ^2 σϵ^2 /. InvertibleSolution /. nValues, {λ > 0, σu^2 > 0, c > 0}]
```

$\frac{1}{2}\left(5 + \sigma_u^2 - \sqrt{(1 + \sigma_u^2)(9 + \sigma_u^2)}\right)$

```
Simplify[θ^2 σϵ^2 /. InvertibleSolution /. nValues /. σu^2 → 0, {λ > 0, σu^2 > 0, c > 0}]
```

1

```
Solve[(θ^2 σϵ^2 /. InvertibleSolution) == (c - λ)^2 + σu^2, σu^2]
```

$\{\{\sigma_u^2 \to 0\}\}$

```
InvertibleSolution /. nValues /. σu^2 → 0 // Simplify
```

$\{\sigma_\epsilon^2 \to 4,\ \theta \to -\frac{1}{2}\}$

Exercise 9.1 (Glosten and Harris, Source notebook: MultivariateMicrostructureModels)

Definitions:

```
mRule = m_t_ :> m_{t-1} + w_t;
wRule = w_t_ :> λ0 q_t + λ1 Q_t + u_t;
pRule = p_t_ :> m_t + c1 q_t + c2 Q_t;
```

With these definitions, m_t, w_t and p_t are (respectively):

```
TableForm[{ m_t /. mRule, w_t /. wRule, p_t /. pRule}]
```

$m_{-1+t} + w_t$
$u_t + q_t \lambda_0 + Q_t \lambda_1$
$m_t + c_1 q_t + c_2 Q_t$

The price change at time t is Δp_t =

```
ΔpRule = Δp_t_ :> (p_t /. pRule /. mRule /. wRule) - (p_{t-1} /. pRule);
Simplify[Δp_t /. ΔpRule]
```

$c_1 (-q_{-1+t} + q_t) + c_2 (-Q_{-1+t} + Q_t) + u_t + q_t \lambda_0 + Q_t \lambda_1$

The vector of variables is:

```
yRule = y_t_ :→ Transpose[{{Δp_t, q_t, Q_t}}]; MatrixForm[y_t /. yRule]
```

$$\begin{pmatrix} \Delta p_t \\ q_t \\ Q_t \end{pmatrix}$$

and the disturbances are ϵ_t =

```
εRule = ε_t_ :→ Transpose[{{u_t, q_t, Q_t}}]; MatrixForm[ε_t /. εRule]
```

$$\begin{pmatrix} u_t \\ q_t \\ Q_t \end{pmatrix}$$

With substitutions, the vector of system variables is:

```
(y_t /. yRule /. ΔpRule) // MatrixForm // Simplify
```

$$\begin{pmatrix} c_1 \left(-q_{-1+t} + q_t\right) + c_2 \left(-Q_{-1+t} + Q_t\right) + u_t + q_t \lambda_0 + Q_t \lambda_1 \\ q_t \\ Q_t \end{pmatrix}$$

The MA model is $y_t = \Theta_0 \epsilon_t + \Theta_1 \epsilon_{t-1}$ where the coefficient matrices are:

```
ClearAll[θ];
θRules = Table[Rule[θ_s, Table[Coefficient[(y_t /. yRule /. ΔpRule)[[i, 1]],
      (ε_{t-s} /. εRule)[[j, 1]]], {i, 3}, {j, 3}]], {s, 0, 1}];
{Map[MatrixForm, θRules, {2}]}
```

$$\left\{\left\{\theta_0 \to \begin{pmatrix} 1 & c_1 + \lambda_0 & c_2 + \lambda_1 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \theta_1 \to \begin{pmatrix} 0 & -c_1 & -c_2 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}\right\}\right\}$$

To build the covariance matrix, we need to work out $\text{Cov}(q_t, Q_t) = E[Q_t \,\text{Sign}(Q_t)] = E\,|\,Q_t\,|$

```
ClearAll[Ω]
```

If $Q_t \sim N\left(0, \sigma_Q^2\right)$, then $E[\,|\,Q\,|\,]$ =

```
Simplify[∫_{-∞}^{∞} Abs[x] PDF[NormalDistribution[0, σ_Q], x] ⅆx, {σ_Q ∈ Reals, Re[(σ_Q)^2] > 0}]
```

$$\sqrt{\frac{2}{\pi}}\, \sigma_Q$$

So the disturbance covariance matrix is:

```
ΩRule = Ω → ({{σu^2, 0, 0}, {0, 1, σq,Q}, {0, σq,Q, σQ^2}} /. σq,Q → Sqrt[2 σQ^2 / π]);
Map[MatrixForm, ΩRule, 1]
```

$$\Omega \to \begin{pmatrix} \sigma_u^2 & 0 & 0 \\ 0 & 1 & \sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2} \\ 0 & \sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2} & \sigma_Q^2 \end{pmatrix}$$

The random-walk variance is σ_w^2 =

```
Simplify[(θ0 + θ1).Ω.Transpose[θ0 + θ1] /. θRules /. ΩRule, {σQ > 0, σ_ ∈ Reals}][[1, 1]]
```

$$\sigma_u^2 + \lambda_0^2 + 2\sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_0\,\lambda_1 + \sigma_Q^2\,\lambda_1^2$$

□ *Decomposition with q_t first:*

```
F1 = Simplify[Permute[CholeskyDecomposition[Permute[Ω /. ΩRule, {2, 3, 1}]],
    {3, 1, 2}], {σQ^2 > 0, σu^2 > 0}];
F1 // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2} & 0 & 0 \\ 0 & 1 & \sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2} \\ 0 & 0 & \sqrt{\frac{-2+\pi}{\pi}}\sqrt{\sigma_Q^2} \end{pmatrix}$$

```
VarDecomp1 = ((θ0 + θ1).Transpose[F1] /. θRules // Simplify)[[1]]
```

$$\left\{\sqrt{\sigma_u^2},\ \lambda_0 + \sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_1,\ \sqrt{\frac{-2+\pi}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_1\right\}$$

The variance components corresponding to u_t, q_t and Q_t are:

```
{VarDecomp1^2 // Simplify} // TableForm
```

$$\sigma_u^2 \quad \left(\lambda_0 + \sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_1\right)^2 \quad \frac{(-2+\pi)\,\sigma_Q^2\,\lambda_1^2}{\pi}$$

Verify that they add up to the correct σ_w^2:

```
Plus @@ VarDecomp1^2 // Simplify
```

$$\sigma_u^2 + \lambda_0^2 + 2\sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_0\,\lambda_1 + \sigma_Q^2\,\lambda_1^2$$

□ *Decomposition with Q_t first:*

```
F2 = Simplify[Permute[CholeskyDecomposition[Permute[Ω /. ΩRule, {3, 2, 1}]],
    {3, 2, 1}], {σQ > 0, σu > 0}];
F2 // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2} & 0 & 0 \\ 0 & \sqrt{\frac{-2+\pi}{\pi}} & 0 \\ 0 & \sqrt{\frac{2}{\pi}} & \sqrt{\sigma_Q^2} \end{pmatrix}$$

```
VarDecomp2 = ((θ0 + θ1).Transpose[F2] /. θRules // Simplify)[[1]]
```

$$\left\{\sqrt{\sigma_u^2},\ \sqrt{\frac{-2+\pi}{\pi}}\,\lambda_0,\ \sqrt{\frac{2}{\pi}}\,\lambda_0 + \sqrt{\sigma_Q^2}\,\lambda_1\right\}$$

```
{VarDecomp2^2} // Simplify // TableForm
```

$$\sigma_u^2 \qquad \frac{(-2+\pi)\,\lambda_0^2}{\pi} \qquad \left(\sqrt{\frac{2}{\pi}}\,\lambda_0 + \sqrt{\sigma_Q^2}\,\lambda_1\right)^2$$

... and verify:

```
Plus @@ VarDecomp2^2 // Simplify
```

$$\sigma_u^2 + \lambda_0^2 + 2\sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_0\,\lambda_1 + \sigma_Q^2\,\lambda_1^2$$

Exercise 9.2 (Madhavan, Richardson and Roomans, Source notebook: MultivariateMicrostructureModels)

The model is:

```
qRule = qt_ :→ vt + β qt-1;
mRule = mt_ :→ mt-1 + wt;
wRule = wt_ :→ λ vt + ut;
pRule = pt_ :→ mt + c qt;
```

With these rules, q_t, m_t, w_t, and p_t are:

```
{qt /. qRule, mt /. mRule, wt /. wRule, pt /. pRule} // TableForm
```

$\beta\, q_{-1+t} + v_t$
$m_{-1+t} + w_t$
$u_t + \lambda\, v_t$
$m_t + c\, q_t$

The price change is $\triangle p_t$=

```
ΔpRule = Δpt_ :> Δpt → (pt /. pRule /. mRule /. wRule) − (pt-1 /. pRule);
Simplify[Δpt /. ΔpRule]
```

$\triangle p_t \to -c\, q_{-1+t} + c\, q_t + u_t + \lambda\, v_t$

The system variables are y_t =

```
yRule = yt_ :> Transpose[{{Δpt, qt}}]; yt /. yRule // MatrixForm
```

$$\begin{pmatrix} \triangle p_t \\ q_t \end{pmatrix}$$

The vector of disturbances is:

```
ϵRule = ϵt_ :> Transpose[{{ut, vt}}]; ϵt /. ϵRule // MatrixForm
```

$$\begin{pmatrix} u_t \\ v_t \end{pmatrix}$$

```
ClearAll[θ]
```

With substitutions, the vector of system variables becomes y_t =

```
{{Δpt}, {qt}} /. Δpt → (pt /. pRule /. mRule /. wRule) − (pt-1 /. pRule) /. qt → (qt /. qRule) //
  Simplify // MatrixForm
```

$$\begin{pmatrix} c\,(-1+\beta)\, q_{-1+t} + u_t + (c+\lambda)\, v_t \\ \beta\, q_{-1+t} + v_t \end{pmatrix}$$

This is a first-order vector autoregressive process: $y_t = \phi\, y_{t-1} + \Theta\, \epsilon_t$ where

```
ϕRule = ϕ → {{0, c (-1 + β)}, {0, β}}; MatrixForm /@ ϕRule
```

$$\phi \to \begin{pmatrix} 0 & c\,(-1+\beta) \\ 0 & \beta \end{pmatrix}$$

and

```
θRule = θ → {{1, (c + λ)}, {0, 1}}; MatrixForm /@ θRule
```

$$\theta \to \begin{pmatrix} 1 & c+\lambda \\ 0 & 1 \end{pmatrix}$$

It may be put in vector moving average (VMA) form as: $y_t = \underbrace{(I - \phi L)^{-1}\theta}_{\text{VMA coefficients}} \epsilon_t$. We could obtain the VMA coefficient matrices by doing the series expansion. Here, though, to compute the random-walk variance, we just need the sum of the moving average coefficients, and $(I - \phi)^{-1}\theta =$

```
maSum = Inverse[IdentityMatrix[2] - (ϕ /. ϕRule)].(θ /. θRule) // Simplify;
maSum // MatrixForm
```

$$\begin{pmatrix} 1 & \lambda \\ 0 & \frac{1}{1-\beta} \end{pmatrix}$$

To compute the random-walk variance σ_w^2, take the upper left hand entry of $\left[(I-\phi)^{-1}\theta\right]\Omega\left[(I-\phi)^{-1}\theta\right]'$

```
maSum.{{σu², 0}, {0, σv²}}.Transpose[maSum] // Simplify // MatrixForm
```

$$\begin{pmatrix} \sigma_u^2 + \lambda^2 \sigma_v^2 & \frac{\lambda \sigma_v^2}{1-\beta} \\ \frac{\lambda \sigma_v^2}{1-\beta} & \frac{\sigma_v^2}{(-1+\beta)^2} \end{pmatrix}$$

The quantity $\sigma_u^2 + \lambda^2 \sigma_v^2$ summarizes the public information and trade-related components of the random-walk.

Exercise 11.1 (Soure notebook: DealersAndInventories)

(The numbers in Exercise 11.1 were used to generate the figure.)

```
f[a_, b_, c_, x_] := a x^2 + b x + c;
```

Inverse arrival rates of sellers (who receive the bid price)

```
ps = f[.2, .2, 2, λ]
```

$$2 + 0.2\lambda + 0.2\lambda^2$$

```
Plot[ps, {λ, 0, 1}]
```



Inverse arrive rate of buyers (who pay the ask)

```
pb = f[.2, -1, 3, λ]
```

$3 - \lambda + 0.2\,\lambda^2$

□ *The single price equilbrium*

```
Plot[{ps, pb}, {λ, 0, 1}, AxesLabel → {"λ", "p"}]
```



Determination of the single-price equilibrium arrival rate:

```
λEq = λ /. Flatten[NSolve[ps == pb, λ]]
```

```
0.833333
```

```
pEq = ps /. λ → λEq
```

```
2.30556
```

(Optimal) average profit and arrival intensity:

```
sol = NMaximize[{λ (pb - ps), λ > 0}, λ]
```

```
{0.208333, {λ → 0.416667}}
```

```
λOpt = λ /. sol[[2, 1]]
```

```
0.416667
```

```
bid = ps /. λ → λOpt
```

```
2.11806
```

```
ask = pb /. λ → λOpt
```

```
2.61806
```

```
extra = {GrayLevel[0.9`], Rectangle[{0.01`, bid}, {λOpt, ask}],
   GrayLevel[0], Dashing[{0.01`}], Line[{{λOpt, 0}, {λOpt, ask}}],
   Line[{{0, bid}, {λOpt, bid}}], Line[{{0, ask}, {λOpt, ask}}],
   Line[{{λEq, 0}, {λEq, pEq}}], Line[{{0, pEq}, {λEq, pEq}}],
   Text["λ^Buy(p)", {0.3`, 2.9`}],
   Text["λ^Sell(p)", {1.05`, 2.6`}]
  };
Plot[{ps, pb}, {λ, 0, 1.2`}, PlotRange → {{0, 1.2`}, {1.8`, 3.1`}},
 Frame → True, FrameLabel → {"Arrival rate", "Price, p"},
 Background → GrayLevel[1], FrameTicks →
  {{{λOpt, "λ^Optimal "}, {λEq, "λ^Eq"}}, {{bid, Bid}, {ask, Ask}, {pEq, "p^Eq"}},
   None, None}, Epilog → extra, BaseStyle → baseStyle]
```



Exercise 15.1 (Source notebook: TradingStrategiesI)

Model dynamics

```
mRule = m_t_ :→ (m_{t-1} /. m_0 → 0) + λ_t s_t + μ + ϵ_t;
pRule = p_t_ :→ m_t + γ (s_t /. s_0 → 0);
WRule = W_t_ :→ W_{t-1} − (s_{t-1} /. s_0 → 0) /; ! t === 0;
ϵZap = ϵ_ → 0;
```

```
solution = OptOrders[3, {pRule, mRule, εZap} /. μ → 0 /. γ → 0] // Simplify;
```

Lagrangian: $\delta\,(1 - s_1 - s_2 - s_3) + s_1^2\,\lambda_1 + s_2\,(s_1\,\lambda_1 + s_2\,\lambda_2) + s_3\,(s_1\,\lambda_1 + s_2\,\lambda_2 + s_3\,\lambda_3)$

First order conditions:

$$-\delta + 2\,s_1\,\lambda_1 + s_2\,\lambda_1 + s_3\,\lambda_1 == 0$$
$$-\delta + s_1\,\lambda_1 + 2\,s_2\,\lambda_2 + s_3\,\lambda_2 == 0$$
$$-\delta + s_1\,\lambda_1 + s_2\,\lambda_2 + 2\,s_3\,\lambda_3 == 0$$
$$1 - s_1 - s_2 - s_3 == 0$$

Solutions:

$$\delta \to \frac{2\,\lambda_1\,(\lambda_2^2 + \lambda_1\,\lambda_3 - 4\,\lambda_2\,\lambda_3)}{\lambda_2\,(\lambda_2 - 4\,\lambda_3)}$$
$$s_1 \to 1 + \frac{2\,\lambda_1\,\lambda_3}{\lambda_2^2 - 4\,\lambda_2\,\lambda_3}$$
$$s_2 \to \frac{\lambda_1\,(\lambda_2 - 2\,\lambda_3)}{\lambda_2\,(\lambda_2 - 4\,\lambda_3)}$$
$$s_3 \to -\frac{\lambda_1}{\lambda_2 - 4\,\lambda_3}$$

```
solution /. {λ1 → 2, λ2 → 1, λ3 → 1 / 2} // TableForm
```

$$\delta \to 0$$
$$s_1 \to -1$$
$$s_2 \to 0$$
$$s_3 \to 2$$

As a check, verify that with constant λ, we obtain the original solution to the basic problem.

```
solution /. {λ_ → λ} // TableForm
```

$$\delta \to \frac{4\,\lambda}{3}$$
$$s_1 \to \frac{1}{3}$$
$$s_2 \to \frac{1}{3}$$
$$s_3 \to \frac{1}{3}$$